

Lesson 10

Integration of Trigonometric Functions

Initializations

```
> restart;  
with(student):  
with(oneonta):  
>
```

10.1 Integration of Trigonometric Functions

Examples

Example 10.1.1

Evaluate $\int \sin^4 x \cos^5 x \, dx$.

Solution

Since one of the exponents is odd, so a simple u -substitution will suffice.

```
> e1:=Int(sin(x)^4*cos(x)^5, x);  
rr:=u=sin(x);
```

$$e1 := \int \sin(x)^4 \cos(x)^5 \, dx$$

$$rr := u = \sin(x) \tag{2.1.1.1}$$

```
> e2:=changevar(rr, e1, u);
```

$$e2 := \int u^4 (1 - u^2)^2 \, du \tag{2.1.1.2}$$

Let Maple evaluate this elementary integral.

```
> e3:=value(e2)+C;
```

$$e3 := \frac{1}{9} u^9 - \frac{2}{7} u^7 + \frac{1}{5} u^5 + C \tag{2.1.1.3}$$

Substitute $u = \sin x$ into the expression above to obtain the answer in terms of the variable x .

```
> e4:=subs(rr, e3);
```

$$e4 := \frac{1}{9} \sin(x)^9 - \frac{2}{7} \sin(x)^7 + \frac{1}{5} \sin(x)^5 + C \tag{2.1.1.4}$$

```
>
```

Example 10.1.2

Evaluate $\int \tan^6 x \sec^4 x dx$.

Solution

Since the power of the secant is even, the substitution $u = \tan x$ will solve the problem.

```
> e1:=Int(tan(x)^6*sec(x)^4, x);  
rr:=u=tan(x);
```

$$e1 := \int \tan(x)^6 \sec(x)^4 dx$$
$$rr := u = \tan(x) \quad (2.1.2.1)$$

```
> e2:=changevar(rr, e1, u);
```

$$e2 := \int u^6 (1 + u^2) du \quad (2.1.2.2)$$

Let Maple evaluate this elementary integral.

```
> e3:=value(e2)+C;
```

$$e3 := \frac{1}{9} u^9 + \frac{1}{7} u^7 + C \quad (2.1.2.3)$$

Substitute $u = \tan x$ into the expression above to obtain the answer in terms of the variable x .

```
> e4:=subs(rr, e3);
```

$$e4 := \frac{1}{9} \tan(x)^9 + \frac{1}{7} \tan(x)^7 + C \quad (2.1.2.4)$$

```
>
```

Example 10.1.3

Evaluate $\int \tan^3 x \sec^5 x dx$.

Solution

Since the power of the tangent is odd and the power of the secant is odd, we have to rely on the substitution $u = \sec x$ to solve the problem.

```
> e1:=Int(tan(x)^3*sec(x)^5, x);  
rr:=u=sec(x);
```

$$e1 := \int \tan(x)^3 \sec(x)^5 dx$$
$$rr := u = \sec(x) \quad (2.1.3.1)$$

```
> e2:=changevar(rr, e1, u);
```

$$e2 := \int u^6 \left(1 - \frac{1}{u^2}\right) du \quad (2.1.3.2)$$

Let Maple evaluate this elementary integral.

```
> e3:=value(e2)+C;
```

$$e3 := \frac{1}{7} u^7 - \frac{1}{5} u^5 + C \quad (2.1.3.3)$$

Substitute $u = \sec x$ into the expression above to obtain the answer in terms of the variable x .

```
> e4:=subs(rr, e3);
```

$$e4 := \frac{1}{7} \sec(x)^7 - \frac{1}{5} \sec(x)^5 + C \quad (2.1.3.4)$$

>

Example 10.1.4

Evaluate $\int \sin^6 4x \, dx$.

Solution

The power of the sine is even, so we have to use a double angle formula. The necessary routine can be found in the **oneonta** package.

> **e1:=Int(sin(4*x)^6, x);**

$$e1 := \int \sin(4x)^6 \, dx \quad (2.1.4.1)$$

> **e2:=subs(doubleangle(4, 6, x), e1);**

$$e2 := \int \left(\frac{1}{2} - \frac{1}{2} \cos(8x) \right)^3 \, dx \quad (2.1.4.2)$$

> **e3:=expand(e2);**

$$e3 := \frac{1}{8} \int 1 \, dx - \frac{3}{8} \int \cos(8x) \, dx + \frac{3}{8} \int \cos(8x)^2 \, dx - \frac{1}{8} \int \cos(8x)^3 \, dx \quad (2.1.4.3)$$

Since there still is one integral with an even power of the cosine function, we apply the **doubleangle** routine once more.

> **e4:=subs(doubleangle(8, 2, x), e3);**

$$e4 := \frac{1}{8} \int 1 \, dx - \frac{3}{8} \int \cos(8x) \, dx + \frac{3}{8} \int \left(\frac{1}{2} + \frac{1}{2} \cos(16x) \right) \, dx - \frac{1}{8} \int \cos(8x)^3 \, dx \quad (2.1.4.4)$$

Now use Maple's integration routine to evaluate this elementary integral.

> **e5:=value(e4)+C;**

$$e5 := \frac{5}{16} x - \frac{11}{192} \sin(8x) + \frac{3}{256} \sin(16x) - \frac{1}{192} \cos(8x)^2 \sin(8x) + C \quad (2.1.4.5)$$

Note that equivalent trigonometric expressions can appear quite different, for example application of Maple's integration routine to the original integral, yields

> **e6:=value(e1)+C;**

$$e6 := -\frac{1}{24} \sin(4x)^5 \cos(4x) - \frac{5}{96} \sin(4x)^3 \cos(4x) - \frac{5}{64} \cos(4x) \sin(4x) + \frac{5}{16} x + C \quad (2.1.4.6)$$

To show that the expressions e5 and e6 are equivalent, use the combine command.

> **check:=combine(e5-e6, trig);**

$$check := 0 \quad (2.1.4.7)$$

>